DerShift

This example demonstrates by a computation over a noncommutative Ore domain how one can use `homalg` to solve a linear system of differential-difference equations:

```
> restart:
> with(JanetOre): with(homalg):
```

Specify the `homalg`-table of the ring package `JanetOre`:

```
> RPO:=`JanetOre/homalg`;
```

$$RPO := JanetOre/homalg$$

Use the ring package `JanetOre` as the default ring package:

```
> `homalg/default`:=RPO;
```

$$homalg/default := JanetOre/homalg$$

Ask homalg to check if a stably free module with a *finite free* resolution[1] is free, cf. [QRar, Remark 51]:

```
> homalg_options("check_rank_1"=true);
```

"JanetOre/homalg", "check_if_stably_free_rank_1_FFR_is_free" = *true*

Define the Ore domain $R = \mathbb{Q}[t][D, t \mapsto t, \frac{d}{dt}][\delta, t \mapsto t+1, 0]$:

```
> Ore:=[[t,D,delta],[],[weyl(D,t),shift(delta,t)]];
```

$$Ore := [[t,\ \mathrm{D},\ \delta],\ [],\ [\mathrm{weyl}(\mathrm{D},\ t),\ \mathrm{shift}(\delta,\ t)]]$$

Find the general solution to the following linear system of differential-difference equations:

$$\begin{aligned} u(t) + u(t+2) + \mathrm{D}(v)(t+1) - v(t+2) + w(t+1) &= 0, \\ t\mathrm{D}(u)(t+1) + u(t+3) - 2\,tu(t+1) + 2\,\mathrm{D}(u)(t+1) - 2\,u(t+1) & \\ -v(t+3) + 2\,v(t+2) + w(t+2) &= 0. \end{aligned}$$

This system can be written as the following matrix operators A applied to the section $\left(\ u(t) \quad v(t) \quad w(t)\ \right)^{\mathrm{tr}}$. (Caution: A monomial in t, D and δ must be read as a monomial with powers of t on the left of a monomial in the commuting D and δ: read in the following both tD, Dt as tD, and both $t\delta$, δt as $t\delta$):

```
> A := matrix([[1+delta^2, delta*D-delta^2, delta],
[t*delta*D+delta^3-2*t*delta+2*delta*D-2*delta, -delta^3+2*delta^2,
delta^2]]);
```

$$A := \begin{bmatrix} 1+\delta^2 & \delta\,\mathrm{D} - \delta^2 & \delta \\ t\,\delta\,\mathrm{D} + \delta^3 - 2\,t\,\delta + 2\,\delta\,\mathrm{D} - 2\,\delta & -\delta^3 + 2\,\delta^2 & \delta^2 \end{bmatrix}$$

Studying the system is equivalent to studying the cokernel M of $A \in R^{m\times n}$ viewed as a morphism $R^{1\times m} \xrightarrow{A} R^{1\times n}$. This is based on the following observation: Let $\mathbb{F}$ be a function space, where one wants to search for solutions of the system, then $\mathrm{Hom}_R(\mathrm{coker}(A), \mathbb{F}) \cong \mathrm{Sol}_{\mathbb{F}}(A) := \{\eta \in \mathbb{F}^{n\times 1} \mid A\eta = 0\}$, where n is the number of columns of A.

```
> M:=Cokernel(A,Ore);
```

$$\begin{aligned} M := [&[[1,\ 0,\ 0] = [1,\ 0,\ 0],\ [0,\ 1,\ 0] = [0,\ 1,\ 0],\ [0,\ 0,\ 1] = [0,\ 0,\ 1]], \\ &[[1+\delta^2,\ \delta\,\mathrm{D} - \delta^2,\ \delta],\ [t\,\delta\,\mathrm{D} + \delta^3 - 2\,t\,\delta + 2\,\delta\,\mathrm{D} - 2\,\delta,\ -\delta^3 + 2\,\delta^2,\ \delta^2]], \\ &\text{“Presentation”},\ 3 + 9\,s + 17\,s^2 + s^3\,(\frac{17}{1-s} + \frac{8}{(1-s)^2} + \frac{1}{(1-s)^3}),\ [17,\ 8,\ 1]] \end{aligned}$$

The torsion free factor F of M turns out to be free of rank 1 (the above mentioned option `"check_rank_1"` was used):

```
> F:=TorsionFreeFactor(M,Ore);
```

$$F := [[1 = [-\delta,\ \delta - \mathrm{D},\ -1]],\ [0],\ \text{“Presentation”},\ \frac{1}{(1-s)^3},\ [0,\ 0,\ 1]]$$

[1] Without a finite free resolution `homalg` cannot check stably freeness, cf. [QRar, Remark 28].

The natural epimorphsim $M \xrightarrow{\nu} F$:

```
> nu:=TorsionFreeFactorEpi(M,Ore);
```

$$\nu := \begin{bmatrix} \delta \\ t \\ -2 + \delta + t\,\delta - t\,\mathrm{D} - \delta^2 \end{bmatrix}$$

Since F is free one can use `Leftinverse` to compute a split $M \xrightarrow{\chi} F$ (cf. [BR, 3.1.3,(1)]):

```
> chi:=Leftinverse(M,nu,F,Ore);
```

$$\chi := \begin{bmatrix} -\delta & \delta - \mathrm{D} & -1 \end{bmatrix}$$

Now compute the torsion submodule T of M:

```
> T:=TorsionSubmodule(M,Ore);
```

$$T := [[[1,\, 0] = [t+1,\, -\delta,\, 0],\, [0,\, 1] = [-2\,\delta,\, -t\,\delta + t\,\mathrm{D} + \delta^2 + 1,\, t]],$$
$$[[1+\delta^2,\, \delta],\, [-2+\mathrm{D},\, -2\,\delta + \delta\,\mathrm{D}],\, [-2\,\delta + \delta\,\mathrm{D},\, 0]],\, \text{“Presentation”},$$
$$2 + 6\,s + s^2\,(\frac{6}{1-s} + \frac{3}{(1-s)^2}),\, [6,\, 3,\, 0]]$$

And the natural embedding $T \xrightarrow{\iota} M$:

```
> iota:=TorsionSubmoduleEmb(M,Ore);
```

$$\iota := \begin{bmatrix} t+1 & -\delta & 0 \\ -2\,\delta & -t\,\delta + t\,\mathrm{D} + \delta^2 + 1 & t \end{bmatrix}$$

After a few tests we found an element $\mu \in T$

```
> mu:=[[delta,1]];
```

$$\mu := [[\delta,\, 1]]$$

which turned out to be a cyclic generator of T (below one identifies μ with the morphism $R^{1\times 1} \rightarrow T:\ 1 \mapsto \mu$):

```
> IsSurjective(mu,T,Ore);
```

true

H is nothing but T rewritten on this cyclic generator. It further turns out that the cyclic generator μ of the torsion submodule T satisfies a single simple relation $\delta^2(D-2)\mu = 0$:

```
> H:=Image(mu,T,Ore);
```

$$H := [[1 = [t\,\delta,\, t\,\mathrm{D} - t\,\delta + 1,\, t]],\, [-2\,\delta^2 + \delta^2\,\mathrm{D}],\ \text{“Presentation”},$$
$$1 + 3\,s + 6\,s^2 + s^3\,(\frac{6}{1-s} + \frac{3}{(1-s)^2}),\, [6,\, 3,\, 0]]$$

ε is nothing but ι rewritten on the cyclic generator μ:

```
> epsilon:=ComposeMaps(mu,iota,M,Ore);
```

$$\varepsilon := \begin{bmatrix} t\,\delta & t\,\mathrm{D} - t\,\delta + 1 & t \end{bmatrix}$$

```
> IsHom(H,epsilon,M,Ore);
```

true

```
> IsInjective(H,epsilon,M,Ore);
```

true

Now construct the isomorphism α from the direct sum $S := F \oplus H$ onto M:

```
> alpha:=RPO[matrix](RPO[UnionOfRows](chi,epsilon));
```

$$\alpha := \begin{bmatrix} -\delta & \delta - \mathrm{D} & -1 \\ t\,\delta & t\,\mathrm{D} - t\,\delta + 1 & t \end{bmatrix}$$

```
> S:=DirectSum(F,H,Ore);
```

$$S := [[[1,\ 0] = [1,\ 0],\ [0,\ 1] = [0,\ 1]],\ [[0,\ -2\,\delta^2 + \delta^2\,\mathrm{D}]],\ \text{“Presentation”}\ ,$$
$$2 + 6\,s + 12\,s^2 + s^3\,(\frac{12}{1-s} + \frac{6}{(1-s)^2} + \frac{1}{(1-s)^3}),\ [12,\ 6,\ 1]]$$

```
> IsHom(S,alpha,M,Ore);
```

$$\mathit{true}$$

```
> IsBijective(S,alpha,M,Ore);
```

$$\mathit{true}$$

N is now nothing but M rewritten on the images of the two generators of S. So M (or N) is a module generated by two generators, one free and one subject to a single simple relation:

```
> N:=Image(alpha,M,Ore);
```

$$N := [[[1,\ 0] = [-\delta,\ \delta - \mathrm{D},\ -1],\ [0,\ 1] = [t\,\delta,\ t\,\mathrm{D} - t\,\delta + 1,\ t]],\ [[0,\ -2\,\delta^2 + \delta^2\,\mathrm{D}]],$$
$$\text{“Presentation”},\ 2 + 6\,s + 12\,s^2 + s^3\,(\frac{12}{1-s} + \frac{6}{(1-s)^2} + \frac{1}{(1-s)^3}),\ [12,\ 6,\ 1]]$$

Using sections, the above presentation reads: The original system on the three unkown functions $u(t)$, $v(t)$ and $w(t)$ is equivalent to a system on two unknown functions $f(t)$ and $g(t)$, where only $g(t)$ satisfies the single simple equation $D(g)(t+1) - g(t+1) = 0$. It is also given how to express f and g in terms of u, v and w.

```
> PresentationOnSections(N,Ore[1],Ore[3],[u,v,w],[f,g]);
```

$$[[\mathrm{f}(t) = -\mathrm{u}(t+1) + \mathrm{v}(t+1) - \mathrm{D}(v)(t) - \mathrm{w}(t),$$
$$\mathrm{g}(t) = t\,\mathrm{u}(t+1) + t\,\mathrm{D}(v)(t) - t\,\mathrm{v}(t+1) + \mathrm{v}(t) + t\,\mathrm{w}(t)],\ [-2\,\mathrm{g}(t+2) + \mathrm{D}(g)(t+2)],$$
$$\text{“Presentation”},\ 2 + 6\,s + 12\,s^2 + s^3\,(\frac{12}{1-s} + \frac{6}{(1-s)^2} + \frac{1}{(1-s)^3}),\ [12,\ 6,\ 1]]$$

$f(t)$ is therefore a free function and $g(t) = C\exp(2t)$, where C is an arbitrary constant:

```
> sol:=[f=unapply(f(t),t),g=unapply(C*exp(2*t),t)];
```

$$\mathit{sol} := [f = (t \to \mathrm{f}(t)),\ g = (t \to C\,e^{(2\,t)})]$$

In order to solve the original system one needs to express u, v and w in terms of f and g. To this end use `Leftinverse` to compute $\eta := \alpha^{-1}$ (cf. [BR, 3.1.3,(2)]):

```
> eta:=Leftinverse(S,alpha,M,Ore);
```

$$\eta := \begin{bmatrix} \delta & 0 \\ t & 1 \\ -2 + \delta + t\,\delta - t\,\mathrm{D} - \delta^2 & \delta - \mathrm{D} \end{bmatrix}$$

Again, L is nothing but S ($\cong M$) rewritten on the images (under η) of the three original generators of M:

```
> L:=Image(eta,S,Ore,"USE_IMAGE_OF_GENERATORS");
```

$$L := [[[1,\ 0,\ 0] = [\delta,\ 0],\ [0,\ 1,\ 0] = [t,\ 1],\ [0,\ 0,\ 1] = [-2 + \delta + t\,\delta - t\,\mathrm{D} - \delta^2,\ \delta - \mathrm{D}]],$$
$$[[1 + \delta^2,\ \delta\,\mathrm{D} - \delta^2,\ \delta],\ [t\,\delta\,\mathrm{D} + \delta^3 - 2\,t\,\delta + 2\,\delta\,\mathrm{D} - 2\,\delta,\ -\delta^3 + 2\,\delta^2,\ \delta^2]],$$
$$\text{“Presentation”},\ 3 + 9\,s + 17\,s^2 + s^3\,(\frac{17}{1-s} + \frac{8}{(1-s)^2} + \frac{1}{(1-s)^3}),\ [17,\ 8,\ 1]]$$

And again, using the `JanetOre` procedure `PresentationOnSections`, the above presentation is rewritten:

```
> P:=PresentationOnSections(L,Ore[1],Ore[3],[f,g],[u,v,w]);
```

$$P := [[\mathrm{u}(t) = \mathrm{f}(t+1),\ \mathrm{v}(t) = t\,\mathrm{f}(t) + \mathrm{g}(t),$$
$$\mathrm{w}(t) = -2\,\mathrm{f}(t) + \mathrm{f}(t+1) + t\,\mathrm{f}(t+1) - t\,\mathrm{D}(f)(t) - \mathrm{f}(t+2) + \mathrm{g}(t+1) - \mathrm{D}(g)(t)], [$$
$$\mathrm{u}(t) + \mathrm{u}(t+2) + \mathrm{D}(v)(t+1) - \mathrm{v}(t+2) + \mathrm{w}(t+1), t\,\mathrm{D}(u)(t+1) + \mathrm{u}(t+3)$$
$$-\,2\,t\,\mathrm{u}(t+1) + 2\,\mathrm{D}(u)(t+1) - 2\,\mathrm{u}(t+1) - \mathrm{v}(t+3) + 2\,\mathrm{v}(t+2) + \mathrm{w}(t+2)],$$
$$\text{“Presentation”},\ 3 + 9\,s + 17\,s^2 + s^3\,(\frac{17}{1-s} + \frac{8}{(1-s)^2} + \frac{1}{(1-s)^3}),\ [17,\ 8,\ 1]]$$

The most general $C^\infty(\mathbb{R})$-solution is given by:

```
> Sol:=eval(RPO[matrix](GeneratorsOfPresentation(P)),sol);
```

$$Sol := \begin{bmatrix} \mathrm{f}(t+1) \\ t\,\mathrm{f}(t) + C\,e^{(2\,t)} \\ -2\,\mathrm{f}(t) + \mathrm{f}(t+1) + t\,\mathrm{f}(t+1) - t\,\mathrm{D}(f)(t) - \mathrm{f}(t+2) + C\,e^{(2\,t+2)} - 2\,C\,e^{(2\,t)} \end{bmatrix}$$

Besides, one has the following uniqueness property: If $\widetilde{f}$ and $\widetilde{C}$ lead to the same solution then $\widetilde{f} = f$ and $\widetilde{C} = C$.

Test the general solution with the `JanetOre` procedure `JApplyMatrix`:

```
> JApplyMatrix(A,Sol,Ore[1],Ore[3]);
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

Author: Mohamed Barakat and Daniel Robertz

Date:	2007-06-05
Last modified:	2007-11-13 11:58

References

[BR] Mohamed Barakat and Daniel Robertz, **homalg** – *A meta-package for homological algebra*, submitted. (arXiv:math.AC/0701146 and http://wwwb.math.rwth-aachen.de/homalg). 2, 3

[BR07] ———, **homalg** *project*, 2003-2007, (http://wwwb.math.rwth-aachen.de/homalg).

[QRar] A. Quadrat and D. Robertz, *Computation of projective dimension and bases over the Weyl algebras.* 1

[Rob06] D. Robertz, *Formal Computational Methods for Control Theory*, Ph.D. thesis, RWTH Aachen, Germany, June 2006, This thesis is available at http://darwin.bth.rwth-aachen.de/opus/volltexte/2006/1586.

[Rob07] Daniel Robertz, **JanetOre***: A* Maple *package to compute a* Janet *basis for modules over* Ore *algebras*, 2003-2007, (http://wwwb.math.rwth-aachen.de:8040).

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: mohamed.barakat@rwth-aachen.de

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: daniel@momo.math.rwth-aachen.de